Example 8.13 Heat Conduction in a Slab with Radiation Boundary Conditions
Equation (8.1.29) is solved in Maple below:

```
> restart:with(inttrans):with(plots):
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t}\, u(x,t) = \frac{\partial^2}{\partial x^2}\, u(x,t) \quad (1)$$

```
> u(x,0):=1;
```

$$u(x,0) := 1 \quad (2)$$

```
> bc1:=diff(u(x,t),x)=0;
```

$$bc1 := \frac{\partial}{\partial x}\, u(x,t) = 0 \quad (3)$$

```
> bc2:=diff(u(x,t),x)+u(x,t)=0;
```

$$bc2 := \frac{\partial}{\partial x}\, u(x,t) + u(x,t) = 0 \quad (4)$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) - 1 = \frac{d^2}{dx^2}\, U(x) \quad (5)$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := \frac{d}{dx}\, U(x) = 0 \quad (6)$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := \frac{d}{dx}\, U(x) + U(x) = 0 \quad (7)$$

```
> dsolve(eqs,U(x));
```

$$U(x) = e^{\sqrt{s}\, x}\, \_C2 + e^{-\sqrt{s}\, x}\, \_C1 + \frac{1}{s} \quad (8)$$

```
> U(x):=c[1]*cosh(s^(1/2)*x)+c[2]*sinh(s^(1/2)*x)+1/s;
```

$$U(x) := c_1 \cosh\left(\sqrt{s}\, x\right) + c_2 \sinh\left(\sqrt{s}\, x\right) + \frac{1}{s} \quad (9)$$

```
> eq0:=eval(subs(x=0,bc1)):
> eq1:=eval(subs(x=1,bc2)):
> con:=solve({eq0,eq1},{c[1],c[2]}):
> U(x):=subs(con,U(x)):
> U(x):=factor(simplify(U(x)));
```

(10)

$$U(x) := -\frac{\cosh(\sqrt{s}\,x) - \sinh(\sqrt{s})\sqrt{s} - \cosh(\sqrt{s})}{s\left(\sinh(\sqrt{s})\sqrt{s} + \cosh(\sqrt{s})\right)} \tag{10}$$

```
> P(s):=numer(U(x));
```

$$P(s) := -\cosh(\sqrt{s}\,x) + \sinh(\sqrt{s})\sqrt{s} + \cosh(\sqrt{s}) \tag{11}$$

```
> Q(s):=denom(U(x));
```

$$Q(s) := s\left(\sinh(\sqrt{s})\sqrt{s} + \cosh(\sqrt{s})\right) \tag{12}$$

Maple cannot find the eigenvalues directly.

```
> solve(Q(s),s);
```

$$0, \mathit{RootOf}\left(\_Z\,(\mathrm{e}^{\_Z})^2 - \_Z + (\mathrm{e}^{\_Z})^2 + 1\right)^2 \tag{13}$$

```
> eig:=sinh(s^(1/2))*s^(1/2)+cosh(s^(1/2));
```

$$eig := \sinh(\sqrt{s})\sqrt{s} + \cosh(\sqrt{s}) \tag{14}$$

For convenience s=-λ_2 is substituted to find the eigenvalues.

```
> eiglambda:=simplify(subs(s^(1/2)=I*lambda,s=-lambda^2,eig));
```

$$eiglambda := -\sin(\lambda)\,\lambda + \cos(\lambda) \tag{15}$$

```
> plot(eiglambda,lambda=0..20,thickness=3,title="Figure 8.23.",
  axes=boxed);
```

Figure 8.23.



The roots are:

```
> 0,0,-lambda^2;
```

$$0, 0, -\lambda^2 \tag{16}$$

```
> fsolve(eiglambda,lambda=1);
```

$$0.8603335890 \tag{17}$$

The first 20 eigenvalues are obtained numerically.

```
> N:=20;
```

$$N := 20 \tag{18}$$

```
> l[1]:=fsolve(eiglambda,lambda=0..2);
```

$$l_1 := 0.8603335890 \tag{19}$$

```
> for i from 2 to N do l[i]:=fsolve(eiglambda,lambda=l[i-1]..l
  [i-1]+4);od:
> seq(l[i],i=1..N);
```

0.8603335890, 3.425618459, 6.437298179, 9.529334405, 12.64528722, 15.77128487, 18.90240996, 22.03649673, 25.17244633, 28.30964285, 31.44771464, 34.58642422, 37.72561283, 40.86517033, 44.00501792, 47.14509774, 50.28536634, 53.42579048, 56.56634428, 59.70700731 **(20)**

```
> mu0:=0;
```

$$\mu 0 := 0 \qquad (21)$$

```
> b[2]:=(s-mu0)^2*P(s)/Q(s);
```

$$b_2 := \frac{s\left(-\cosh(\sqrt{s}\,x) + \sinh(\sqrt{s})\sqrt{s} + \cosh(\sqrt{s})\right)}{\sinh(\sqrt{s})\sqrt{s} + \cosh(\sqrt{s})} \qquad (22)$$

```
> B[2]:=limit(b[2],s=0);
```

$$B_2 := 0 \qquad (23)$$

```
> b[1]:=diff(b[2],s):
> B[1]:=limit(b[1],s=0);
```

$$B_1 := 0 \qquad (24)$$

```
> A(s):=P(s)/diff(Q(s),s):
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := -\frac{2\left(\cosh(\sqrt{\mu}\,x) - \sinh(\sqrt{\mu})\sqrt{\mu} - \cosh(\sqrt{\mu})\right)\sqrt{\mu}}{4\sinh(\sqrt{\mu})\,\mu + 2\cosh(\sqrt{\mu})\sqrt{\mu} + \mu^{3/2}\cosh(\sqrt{\mu})} \qquad (25)$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*lambda,mu^(3/2)=-I*lambda^3,mu=-
  lambda^2,A[n]));
```

$$A_n := \frac{2\left(\cos(\lambda x) + \sin(\lambda)\,\lambda - \cos(\lambda)\right)}{4\sin(\lambda)\,\lambda - 2\cos(\lambda) + \lambda^2\cos(\lambda)} \qquad (26)$$

```
> vars:={cos(lambda)=lambda*sin(lambda)};
```

$$vars := \{\cos(\lambda) = \sin(\lambda)\,\lambda\} \qquad (27)$$

```
> A[n]:=simplify(subs(vars,expand(A[n])));
```

$$A_n := \frac{2\cos(\lambda x)}{\sin(\lambda)\,\lambda\,(2+\lambda^2)} \qquad (28)$$

```
> b1s:=B[1]*subs(mu0=0,1/(s-mu0));
```

$$b1s := 0 \qquad (29)$$

```
> b1t:=invlaplace(b1s,s,t);
```

$$b1t := 0 \qquad (30)$$

```
> b2s:=B[2]*subs(mu0=0,1/(s-mu0)^2);
```

$$b2s := 0 \tag{31}$$

```
> b2t:=invlaplace(b2s,s,t);
```

$$b2t := 0 \tag{32}$$

```
> uns:=A[n]/(s-mu);
```

$$uns := \frac{2\cos(\lambda x)}{\sin(\lambda)\,\lambda\,(2+\lambda^2)\,(s-\mu)} \tag{33}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := \frac{2\cos(\lambda x)\,e^{\mu t}}{\sin(\lambda)\,\lambda\,(2+\lambda^2)} \tag{34}$$

```
> unt:=subs(mu=-l[n]^2,lambda=l[n],unt);
```

$$unt := \frac{2\cos(l_n x)\,e^{-l_n^2 t}}{\sin(l_n)\,l_n\,(2+l_n^2)} \tag{35}$$

The solution obtained can then be plotted.

```
> U:=b1t+b2t+Sum(unt,n=1..infinity);
```

$$U := \sum_{n=1}^{\infty} \frac{2\cos(l_n x)\,e^{-l_n^2 t}}{\sin(l_n)\,l_n\,(2+l_n^2)} \tag{36}$$

```
> u:=piecewise(t=0,1,t>0,subs(infinity=20,U)):
> u:=evalf(u):
> plot3d(u,x=0..1,t=0..0.5,axes=boxed,title="Figure 8.24.",labels=
  [x,t,"u"],orientation=[45,60]);
```

Figure 8.24.




```
> plot([subs(t=0,u),subs(t=0.1,u),subs(t=0.2,u),subs(t=0.5,u)],x=
  0..1,title="Figure 8.25.",axes=boxed,thickness=5,labels=[x,"u"])
  ;
```

Figure 8.25.